Techniques of Water-Resources Investigations of the United States Geological Survey

Chapter A1

A MODULAR THREE-DIMENSIONAL FINITE-DIFFERENCE GROUND-WATER FLOW MODEL

By Michael G. McDonald and Arlen W. Harbaugh

This chapter supersedes U.S. Geological Survey Open-File Report 83-875

Book 6

MODELING TECHNIQUES

Click here to return to USGS Publications

Most of the data submitted by the user will consist of one-dimensional and two-dimensional arrays. Those arrays are submitted as an "array control record" plus, optionally, a series of records containing the array elements. The array control record is read from the unit number specified for the major option which calls for the array. If all the elements of an array have the same value, the value is specified on the control record and it is not necessary to read the associated array. If the elements of the array vary, records containing the array values are read from the unit specified on the array control record according to a format which is also specified in the control record. The unit number may be the same as that from which the control record is read, or it may be different. Thus there is a great deal of flexibility regarding the organization of the input data for a simulation.

Any consistent length and time units may be used for model data. This gives a certain amount of freedom to the user, but care must be exercised to avoid any mixing of units. There is no way for the program to detect the use of inconsistent units. For example, if transmissivity is entered in units of ft^2/day and pumpage as m^3/s, the program will run, but the results will be meaningless.

Output Structure

The output structure is designed to control the amount, type, and frequency of information to be printed or written on disk. It controls the printing of head and drawdown by layer and time step, and the printing of the overall volumetric budget. It also controls disk output of head, drawdown, and cell-by-cell flow terms for use in calculations external to the model, or in user-supplied printing and plotting programs.

Output Control, which is a major option contained within the Basic Package, receives instructions from the user to control the amount and frequency of output. To utilize this option, the user must specify the unit number of the file or channel from which the input data for the Output Control option are to be read. This unit number must be entered as the twelfth element of the IUNIT array (IUNIT 12); the input information is then read, at each time step, from the file identified by this unit number. If a zero is specified as the twelfth element of the IUNIT array, a default output convention is invoked. This default output consists of head values and budget terms printed for the end of each stress period. Every simulation generates some printer output. All printer output goes to unit number 6 as specified in the main program. This unit number can be changed to meet the requirements of a particular computer.

The Main Program

The main program serves two major purposes: (1) it controls the order in which the primary modules are executed, and (2) it serves as a switching system for information. It does so with CALL statements which specify, by name, a module to be executed and lists the names of data fields (subroutine arguments) which are accessible by both the main program and the module.

The arrangement of CALL statements in the program reflects the order of procedures shown in the system flow chart (figure 13). Within a procedure, the calls to specific modules can be in any order with one exception: if a procedure has a CALL to a module in the Basic Package, that CALL must precede all other CALLS in that procedure. The main program calls modules to perform the following tasks, in order (the numbers in the following list correspond to the numbers of the comments in the main program listing).

1. Set the length of the "X" array (LENX) in which all data arrays and lists are stored. Note: LENX should be set equal to the dimension of the X array prior to compilation.

2. Assign the input for the Basic Package to unit 1; assign printed output to unit 6.

3. Define the problem in terms of number of rows, columns, layers, stress periods, and major options to be used.

4. Allocate space in the X array for individual data arrays and lists.

5. If the X array is not big enough for the problem, STOP. (Redimension X and redefine LENX.)

6. Read and prepare information which is constant throughout the simulation.

7. For each stress period:

    (a) Read stress-period timing information.

    (b) Read and prepare information that changes each stress period.

    (c) For each time step:

(1) Calculate the current time-step length and move "new" heads from the preceding time step to the array containing "old" heads of the current time step.

(2) Iteratively formulate and solve the system of equations:

a. Formulate the finite-difference equations.

b. Calculate an approximate solution to the system of equations.

c. If convergence criterion has been met, stop iterating.

(3) Determine the type and amount of output needed for this time step.

(4) Calculate overall budget terms and, if specified, calculate and print or record cell-by-cell flow terms.

(5) Print and/or record heads and/or drawdown. Print the overall volumetric budget and timing summary.

(6) If iteration fails to meet convergence criterion, STOP.

8. END PROGRAM.

```
C     ****************************************************************
C     MAIN CODE FOR MODULAR MODEL --    9/1/87
C          BY MICHAEL G. MCDONALD AND ARLEN W. HARBAUGH
C-----VERSION 1638 24JUL1987 MAIN1
C     ****************************************************************
C
C        SPECIFICATIONS:
C     ----------------------------------------------------------------
      COMMON X(30000)
      COMMON /FLWCOM/LAYCON(80)
      CHARACTER*4 HEADNG,VBNM
      DIMENSION HEADNG(32),VBNM(4,20),VBVL(4,20),IUNIT(24)
      DOUBLE PRECISION DUMMY
      EQUIVALENCE (DUMMY,X(1))
C     ----------------------------------------------------------------
C
C1------SET SIZE OF X ARRAY. REMEMBER TO REDIMENSION X.
      LENX=30000
C
C2------ASSIGN BASIC INPUT UNIT AND PRINTER UNIT.
      INBAS=1
      IOUT=6
C
C3------DEFINE PROBLEM__ROWS,COLUMNS,LAYERS,STRESS PERIODS,PACKAGES
      CALL BAS1DF(ISUM,HEADNG,NPER,ITMUNI,TOTIM,NCOL,NROW,NLAY,
     1             NODES,INBAS,IOUT,IUNIT)
C
C4------ALLOCATE SPACE IN "X" ARRAY.
      CALL BAS1AL(ISUM,LENX,LCHNEW,LCHOLD,LCIBOU,LCCR,LCCC,LCCV,
     1             LCHCOF,LCRHS,LCDELR,LCDELC,LCSTRT,LCBUFF,LCIOFL,
     2             INBAS,ISTRT,NCOL,NROW,NLAY,IOUT)
      IF(IUNIT(1).GT.0) CALL BCF1AL(ISUM,LENX,LCSC1,LCHY,
     1     LCBOT,LCTOP,LCSC2,LCTRPY,IUNIT(1),ISS,
     2     NCOL,NROW,NLAY,IOUT,IBCFCB)
      IF(IUNIT(2).GT.0) CALL WEL1AL(ISUM,LENX,LCWELL,MXWELL,NWELLS,
     1                 IUNIT(2),IOUT,IWELCB)
      IF(IUNIT(3).GT.0) CALL DRN1AL(ISUM,LENX,LCDRAI,NDRAIN,MXDRN,
     1                 IUNIT(3),IOUT,IDRNCB)
      IF(IUNIT(8).GT.0) CALL RCH1AL(ISUM,LENX,LCIRCH,LCRECH,NRCHOP,
     1            NCOL,NROW,IUNIT(8),IOUT,IRCHCB)
      IF(IUNIT(5).GT.0) CALL EVT1AL(ISUM,LENX,LCIEVT,LCEVTR,LCEXDP,
     1            LCSURF,NCOL,NROW,NEVTOP,IUNIT(5),IOUT,IEVTCB)
      IF(IUNIT(4).GT.0) CALL RIV1AL(ISUM,LENX,LCRIVR,MXRIVR,NRIVER,
     1            IUNIT(4),IOUT,IRIVCB)
      IF(IUNIT(7).GT.0) CALL GHB1AL(ISUM,LENX,LCBNDS,NBOUND,MXBND,
     1            IUNIT(7),IOUT,IGHBCB)
      IF(IUNIT(9).GT.0) CALL SIP1AL(ISUM,LENX,LCEL,LCFL,LCGL,LCV,
     1          LCHDCG,LCLRCH,LCW,MXITER,NPARM,NCOL,NROW,NLAY,
     2          IUNIT(9),IOUT)
      IF(IUNIT(11).GT.0) CALL SOR1AL(ISUM,LENX,LCA,LCRES,LCHDCG,LCLRCH,
     1           LCIEQP,MXITER,NCOL,NLAY,NSLICE,MBW,IUNIT(11),IOUT)
C
C5------IF THE "X" ARRAY IS NOT BIG ENOUGH THEN STOP.
      IF(ISUM-1.GT.LENX) STOP
C
C6------READ AND PREPARE INFORMATION FOR ENTIRE SIMULATION.
      CALL BAS1RP(X(LCIBOU),X(LCHNEW),X(LCSTRT),X(LCHOLD),
     1       ISTRT,INBAS,HEADNG,NCOL,NROW,NLAY,NODES,VBVL,X(LCIOFL),
     2       IUNIT(12),IHEDFM,IDDNFM,IHEDUN,IDDNUN,IOUT)
      IF(IUNIT(1).GT.0) CALL BCF1RP(X(LCIBOU),X(LCHNEW),X(LCSC1),
     1          X(LCHY),X(LCCR),X(LCCC),X(LCCV),X(LCDELR),
     2           X(LCDELC),X(LCBOT),X(LCTOP),X(LCSC2),X(LCTRPY),
     3           IUNIT(1),ISS,NCOL,NROW,NLAY,NODES,IOUT)
```

```
      IF(IUNIT(9).GT.0) CALL SIP1RP(NPARM,MXITER,ACCL,HCLOSE,X(LCW),
     1           IUNIT(9),IPCALC,IPRSIP,IOUT)
      IF(IUNIT(11).GT.0) CALL SOR1RP(MXITER,ACCL,HCLOSE,IUNIT(11),
     1          IPRSOR,IOUT)
C
C7------SIMULATE EACH STRESS PERIOD.
      DO 300 KPER=1,NPER
      KKPER=KPER
C
C7A-----READ STRESS PERIOD TIMING INFORMATION.
      CALL BAS1ST(NSTP,DELT,TSMULT,PERTIM,KKPER,INBAS,IOUT)
C
C7B-----READ AND PREPARE INFORMATION FOR STRESS PERIOD.
      IF(IUNIT(2).GT.0) CALL WEL1RP(X(LCWELL),NWELLS,MXWELL,IUNIT(2),
     1             IOUT)
      IF(IUNIT(3).GT.0) CALL DRN1RP(X(LCDRAI),NDRAIN,MXDRN,IUNIT(3),
     1                 IOUT)
      IF(IUNIT(8).GT.0) CALL RCH1RP(NRCHOP,X(LCIRCH),X(LCRECH),
     1            X(LCDELR),X(LCDELC),NROW,NCOL,IUNIT(8),IOUT)
      IF(IUNIT(5).GT.0) CALL EVT1RP(NEVTOP,X(LCIEVT),X(LCEVTR),
     1            X(LCEXDP),X(LCSURF),X(LCDELR),X(LCDELC),NCOL,NROW,
     1            IUNIT(5),IOUT)
      IF(IUNIT(4).GT.0) CALL RIV1RP(X(LCRIVR),NRIVER,MXRIVR,IUNIT(4),
     1            IOUT)
      IF(IUNIT(7).GT.0) CALL GHB1RP(X(LCBNDS),NBOUND,MXBND,IUNIT(7),
     1              IOUT)
C
C7C-----SIMULATE EACH TIME STEP.
      DO 200 KSTP=1,NSTP
      KKSTP=KSTP
C
C7C1----CALCULATE TIME STEP LENGTH. SET HOLD=HNEW..
      CALL BAS1AD(DELT,TSMULT,TOTIM,PERTIM,X(LCHNEW),X(LCHOLD),KKSTP,
     1             NCOL,NROW,NLAY)
C
C7C2----ITERATIVELY FORMULATE AND SOLVE THE EQUATIONS.
      DO 100 KITER=1,MXITER
      KKITER=KITER
C
C7C2A---FORMULATE THE FINITE DIFFERENCE EQUATIONS.
      CALL BAS1FM(X(LCHCOF),X(LCRHS),NODES)
      IF(IUNIT(1).GT.0) CALL BCF1FM(X(LCHCOF),X(LCRHS),X(LCHOLD),
     1          X(LCSC1),X(LCHNEW),X(LCIBOU),X(LCCR),X(LCCC),X(LCCV),
     2          X(LCHY),X(LCTRPY),X(LCBOT),X(LCTOP),X(LCSC2),
     3          X(LCDELR),X(LCDELC),DELT,ISS,KKITER,KKSTP,KKPER,NCOL,
     4          NROW,NLAY,IOUT)
      IF(IUNIT(2).GT.0) CALL WEL1FM(NWELLS,MXWELL,X(LCRHS),X(LCWELL),
     1           X(LCIBOU),NCOL,NROW,NLAY)
      IF(IUNIT(3).GT.0) CALL DRN1FM(NDRAIN,MXDRN,X(LCDRAI),X(LCHNEW),
     1         X(LCHCOF),X(LCRHS),X(LCIBOU),NCOL,NROW,NLAY)
      IF(IUNIT(8).GT.0) CALL RCH1FM(NRCHOP,X(LCIRCH),X(LCRECH),
     1            X(LCRHS),X(LCIBOU),NCOL,NROW,NLAY)
      IF(IUNIT(5).GT.0) CALL EVT1FM(NEVTOP,X(LCIEVT),X(LCEVTR),
     1            X(LCEXDP),X(LCSURF),X(LCRHS),X(LCHCOF),X(LCIBOU),
     1            X(LCHNEW),NCOL,NROW,NLAY)
      IF(IUNIT(4).GT.0) CALL RIV1FM(NRIVER,MXRIVR,X(LCRIVR),X(LCHNEW),
     1            X(LCHCOF),X(LCRHS),X(LCIBOU),NCOL,NROW,NLAY)
      IF(IUNIT(7).GT.0) CALL GHB1FM(NBOUND,MXBND,X(LCBNDS),X(LCHCOF),
     1            X(LCRHS),X(LCIBOU),NCOL,NROW,NLAY)
C
```

```
C7C2B---MAKE ONE CUT AT AN APPROXIMATE SOLUTION.
      IF(IUNIT(9).GT.0) CALL SIP1AP(X(LCHNEW),X(LCIBOU),X(LCCR),X(LCCC),
     1     X(LCCV),X(LCHCOF),X(LCRHS),X(LCEL),X(LCFL),X(LCGL),X(LCV),
     2     X(LCW),X(LCHDCG),X(LCLRCH),NPARM,KKITER,HCLOSE,ACCL,ICNVG,
     3     KKSTP,KKPER,IPCALC,IPRSIP,MXITER,NSTP,NCOL,NROW,NLAY,NODES,
     4     IOUT)
      IF(IUNIT(11).GT.0) CALL SOR1AP(X(LCHNEW),X(LCIBOU),X(LCCR),
     1     X(LCCC),X(LCCV),X(LCHCOF),X(LCRHS),X(LCA),X(LCRES),X(LCIEQP),
     2     X(LCHDCG),X(LCLRCH),KKITER,HCLOSE,ACCL,ICNVG,KKSTP,KKPER,
     3     IPRSOR,MXITER,NSTP,NCOL,NROW,NLAY,NSLICE,MBW,IOUT)
C
C7C2C---IF CONVERGENCE CRITERION HAS BEEN MET STOP ITERATING.
      IF(ICNVG.EQ.1) GO TO 110
  100 CONTINUE
      KITER=MXITER
  110 CONTINUE
C
C7C3----DETERMINE WHICH OUTPUT IS NEEDED.
      CALL BAS1OC(NSTP,KKSTP,ICNVG,X(LCIOFL),NLAY,
     1  IBUDFL,ICBCFL,IHDDFL,IUNIT(12),IOUT)
C
C7C4----CALCULATE BUDGET TERMS. SAVE CELL-BY-CELL FLOW TERMS.
      MSUM=1
      IF(IUNIT(1).GT.0) CALL BCF1BD(VBNM,VBVL,MSUM,X(LCHNEW),
     1     X(LCIBOU),X(LCHOLD),X(LCSC1),X(LCCR),X(LCCC),X(LCCV),
     2     X(LCTOP),X(LCSC2),DELT,ISS,NCOL,NROW,NLAY,KKSTP,KKPER,
     3     IBCFCB,ICBCFL,X(LCBUFF),IOUT)
      IF(IUNIT(2).GT.0) CALL WEL1BD(NWELLS,MXWELL,VBNM,VBVL,MSUM,
     1     X(LCWELL),X(LCIBOU),DELT,NCOL,NROW,NLAY,KKSTP,KKPER,IWELCB,
     1     ICBCFL,X(LCBUFF),IOUT)
      IF(IUNIT(3).GT.0) CALL DRN1BD(NDRAIN,MXDRN,VBNM,VBVL,MSUM,
     1     X(LCDRAI),DELT,X(LCHNEW),NCOL,NROW,NLAY,X(LCIBOU),KKSTP,
     2     KKPER,IDRNCB,ICBCFL,X(LCBUFF),IOUT)
      IF(IUNIT(8).GT.0) CALL RCH1BD(NRCHOP,X(LCIRCH),X(LCRECH),
     1     X(LCIBOU),NROW,NCOL,NLAY,DELT,VBVL,VBNM,MSUM,KKSTP,KKPER,
     2     IRCHCB,ICBCFL,X(LCBUFF),IOUT)
      IF(IUNIT(5).GT.0) CALL EVT1BD(NEVTOP,X(LCIEVT),X(LCEVTR),
     1     X(LCEXDP),X(LCSURF),X(LCIBOU),X(LCHNEW),NCOL,NROW,NLAY,
     2     DELT,VBVL,VBNM,MSUM,KKSTP,KKPER,IEVTCB,ICBCFL,X(LCBUFF),IOUT)
      IF(IUNIT(4).GT.0) CALL RIV1BD(NRIVER,MXRIVR,X(LCRIVR),X(LCIBOU),
     1     X(LCHNEW),NCOL,NROW,NLAY,DELT,VBVL,VBNM,MSUM,
     2     KKSTP,KKPER,IRIVCB,ICBCFL,X(LCBUFF),IOUT)
      IF(IUNIT(7).GT.0) CALL GHB1BD(NBOUND,MXBND,VBNM,VBVL,MSUM,
     1     X(LCBNDS),DELT,X(LCHNEW),NCOL,NROW,NLAY,X(LCIBOU),KKSTP,
     2     KKPER,IGHBCB,ICBCFL,X(LCBUFF),IOUT)
C
C7C5---PRINT AND OR SAVE HEADS AND DRAWDOWNS. PRINT OVERALL BUDGET.
      CALL BAS1OT(X(LCHNEW),X(LCSTRT),ISTRT,X(LCBUFF),X(LCIOFL),
     1         MSUM,X(LCIBOU),VBNM,VBVL,KKSTP,KKPER,DELT,
     2         PERTIM,TOTIM,ITMUNI,NCOL,NROW,NLAY,ICNVG,
     3     IHDDFL,IBUDFL,IHEDFM,IHEDUN,IDDNFM,IDDNUN,IOUT)
C
C7C6----IF ITERATION FAILED TO CONVERGE THEN STOP.
      IF(ICNVG.EQ.0) STOP
  200 CONTINUE
  300 CONTINUE
C
C8------END PROGRAM
      STOP
C
      END
```

CHAPTER 4

BASIC PACKAGE

Conceptualization and Implementation

The Basic Package handles a number of administrative tasks for the model. It reads data on the number of rows, columns, layers, and stress periods, on the major options to be used, and on the location of input data for those options. It allocates space in computer memory for model arrays; it reads data specifying initial and boundary conditions; it reads and implements data establishing the discretization of time; it sets up the starting head arrays for each time step; it calculates an overall water budget; and it controls model output according to user specification.

Selection of Major Options and Designation of Input Files

The selection of major options and the designation of their input unit numbers were discussed in the preceding chapter. The primary role of the Basic Package in these operations is to read the IUNIT array; as noted in Chapter 3, the entries in this array determine (a) whether or not a major option is to be used and (b) the unit number from which data for the option is to be read. Whenever a new major option is added to the program, an element corresponding to that option must be added to the IUNIT array.

The IBOUND Array

Recall that the finite-difference equation for a cell has the form

$$CR_{i,j-1/2,k}(h^m_{i,j-1,k}-h^m_{i,j,k}) + CR_{i,j+1/2,k}(h^m_{i,j+1,k} - h^m_{i,j,k})$$
$$+ CC_{i-1/2,j,k}(h^m_{i-1,j,k}-h^m_{i,j,k}) + CC_{i+1/2,j,k}(h^m_{i+1,j,k} - h^m_{i,j,k})$$
$$+ CV_{i,j,k-1/2}(h^m_{i,j,k-1}-h^m_{i,j,k}) + CV_{i,j,k+1/2}(h^m_{i,j,k+1}-h^m_{i,j,k})$$
$$+ P_{i,j,k}h^m_{i,j,k}+Q_{i,j,k} = SC1_{i,j,k}\ (h^m_{i,j,k}-h^{m-1}_{i,j,k})/\Delta t_m. \qquad (28)$$

One equation of this form is written for each variable-head cell in the grid. The IBOUND array, which is specified by the user and read by the Basic Package, contains a code for each cell which indicates whether (1) the head varies with time (variable-head cell), (2) the head is constant (constant-head cell), or (3) no flow takes place within the cell (no-flow or inactive cell). The IBOUND array can be modified by other packages if the state of a cell changes. Figure 19 illustrates the distribution of IBOUND code entries for a typical model layer.

Initial Conditions

Because equation (28) is in backward-difference form, a head distribution at the beginning of each time step is required to calculate the head distribution at the end of that time step (figure 20). For each time step after the first, the head distribution at the start of one time step is set equal to the head distribution at the end of the previous time step. For the first time step, "starting heads" are specified by the user. These specified initial heads are used for head calculation only in the first time step; however, they may also be saved, in the array STRT, and used to



0	1	1	1	1	1	0	0	0	0	0	0	0	1	1	1	1	1	0	0
1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	0
1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	0
0	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	0
0	0	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	0
0	0	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1
0	0	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1
0	0	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	0
0	0	0	0	0	0	1	1	1	1	1	1	1	1	1	1	1	1	0	0
0	0	0	0	0	0	0	-1	-1	-1	-1	-1	-1	-1	-1	-1	-1	0	0	0

IBOUND Codes

< 0 Constant Head
= 0 No Flow
> 0 Variable Head

Figure 19.—Example of the boundary array (IBOUND) for a single layer.

Starting heads (STRT) are the heads at the beginning of the simulation.

New Heads (HNEW) are the latest estimate of the heads at the end of the current time step. Each iteration produces a new estimate.

Old Heads (HOLD) are the heads at the beginning of the current time step. They are, therefore, equal to the heads at the end of the previous time step.



Figure 20.—Flow of head distributions during a simulation.

calculate drawdown, the difference between the starting head distribution and some later head distribution.

Discretization of Time

Simulation time is divided into stress periods--time intervals during which all external stresses are constant--which are, in turn, divided into time steps as shown in figure 21. Within each stress period, the time steps form a geometric progression. The user specifies the length of the stress period, the number of time steps into which it is to be divided, and the time step multiplier, or ratio of the length of each time step to that of the preceding time step. Using these terms, the program calculates the length of each time step in the stress period.

Output

The primary output of the program is head distribution. The user may control the frequency at which heads are printed or saved on disk through the "Output Control" option, a major option contained in the Basic Package. Other output items include drawdowns and volumetric budget terms; the Output Control option also provides for storage or printing of these terms. If Output Control is not utilized, a default output option is invoked--the head distribution and the overall volumetric budget are printed at the end of each stress period, and drawdowns are also printed if starting heads were saved. Figure 22 shows an example of a volumetric budget printout for the end of a stress period.



$$\text{Delt}(1) = \frac{\text{PERLEN} * (1\text{-TSMULT})}{1 - \text{TSMULT} ** \text{NSTP}}$$

$$\text{Delt}(m + 1) = \text{TSMULT} * \text{Delt}(m)$$

Specified by User

PERLEN...........Length of Stress Period
TSMULT ....Time Step Multiplier
NSTP ....Number of Time Steps *in Stress Period*

Calculated by Program

Delt(m)............Length of Time Step m

Figure 21.—Division of simulation time into stress periods and time steps.

```
          VOLUMETRIC BUDGET FOR ENTIRE MODEL AT END OF TIME STEP  1 IN STRESS PERIOD  1
          ------------------------------------------------------------------------------
CUMULATIVE VOLUMES      L**3                                RATES FOR THIS TIME STEP      L**3/T
------------------                                          ------------------------

      IN:                                                          IN:
      ---                                                          ---
        STORAGE =     .0                                             STORAGE =     .0
  CONSTANT HEAD =     .0                                       CONSTANT HEAD =     .0
          WELLS =     .0                                               WELLS =     .0
         DRAINS =     .0                                              DRAINS =     .0
       RECHARGE =     .13608E+08                                    RECHARGE =     157.50
       TOTAL IN =     .13608E+08                                    TOTAL IN =     157.50
     OUT:                                                         OUT:
     ----                                                         ----
        STORAGE =     .0                                             STORAGE =     .0
  CONSTANT HEAD =     .43265E+07                               CONSTANT HEAD =     50.075
          WELLS =     .64800E+07                                       WELLS =     75.000
         DRAINS =     .28010E+07                                      DRAINS =     32.419
       RECHARGE =     .0                                            RECHARGE =     .0
      TOTAL OUT =     .13607E+08                                   TOTAL OUT =     157.49
       IN - OUT =     303.00                                        IN - OUT =     .34943E-02
PERCENT DISCREPANCY =                 0.00                  PERCENT DISCREPANCY =                 0.00
```

Figure 22.--Sample overall volumetric water budget.

Budget Calculations in the Basic Package

The calculation of the volumetric budget is carried out in two parts, the calculation of budget entries and the summation of those entries. As explained in Chapter 3 the entries, which correspond to individual components of flow, are calculated in the flow component packages and stored in the one-dimenisonal array VBVL. The array VBVL is passed to the Basic Package which sums and prints the budget entries.

Basic Package Input

Input for the Basic (BAS) Package except for output control is read from unit 1 as specified in the main program. If necessary, the unit number for BAS input can be changed to meet the requirements of a particular computer. Input for the output control option is read from the unit number specified in IUNIT(12).

Information for the Basic Package must be submitted in the following order:

FOR EACH SIMULATION

BAS1DF

1. Data: HEADNG(32)
   Format: 20A4

2. Data: HEADNG (continued)
   Format: 12A4

3. Data: NLAY NROW NCOL NPER ITMUNI
   Format: I10 I10 I10 I10 I10

4. Data: IUNIT(24)
   Format: 24I3
   (BCF WEL DRN RIV EVT XXX GHB RCH SIP XXX SOR OC)

BAS1AL

5. Data: IAPART ISTRT
   Format: I10 I10

BAS1RP

6. Data: IBOUND(NCOL,NROW)
   Module: U2DINT
   (One array for each layer in the grid)

7. Data: HNOFLO
   Format: F10.0

8. Data: Shead(NCOL,NROW)
   Module: U2DREL
   (One array for each layer in the grid)

NOTE: IBOUND and Shead are treated as three-dimensional arrays in the program. However, the input to each of these arrays is handled as a series of two-dimensional arrays, one for each layer in the grid.

FOR EACH STRESS PERIOD

BAS1ST

9. Data: PERLEN NSTP TSMULT
   Format: F10.0 I10 F10.0

Explanation of Fields Used in
Input Instructions

HEADNG--is the simulation title that is printed on the printout. It may be up to 132 characters long; 80 in the first record and 52 in the second. Both records must be included even if they are blank.

NLAY--is the number of model layers.

NROW--is the number of model rows.

NCOL--is the number of model columns.

NPER--is the number of stress periods in the simulation.

ITMUNI--indicates the time unit of model data. (It is used only for printout of elapsed simulation time. It does not affect model calculations.)

0 - undefined	3 - hours
1 - seconds	4 - days
2 - minutes	5 - years

The unit of time must be consistent for all data values that involve time. For example, if years is the chosen time unit, stress-period length, time-step length, transmissivity, etc., must all be expressed using years for their time units. Likewise, the length unit must also be consistent.

IUNIT--is a 24-element table of input units for use by all major options. Only 10 elements (1-5, 7-9, 11, and 12) are being used. Element 6 has been reserved for a transient leakage package, while element 10 has been reserved for an additional solver, both on the assumption that such packages will be added to the model in the future. Elements 13-24 are reserved for future major options.

IUNIT LOCATION	MAJOR OPTION
1	Block-Centered Flow Package
2	Well Package
3	Drain Package
4	River Package
5	Evapotranspiration Package
6	Reserved for Transient Leakage Package
7	General-Head Boundary Package
8	Recharge Package
9	SIP Package
10	Reserved for additional solver
11	SSOR Package
12	Output Control Option

If IUNIT(n) $\leq$ 0, the corresponding major option is not being used.

If IUNIT(n) > 0, the corresponding major option is being used and data for that option will be read from the unit number contained in IUNIT(n). The unit numbers in IUNIT should be integers from 1 to 99. Although the same number may be used for all or some of the major options, it is recommended that a different number be used for each major option. Printer output is assigned to unit 6 (unless it is changed to meet computer requirements). That unit number should not be used for any other input or output. The user is also permitted to assign unit numbers for output. Those numbers should be different from those assigned to input. The Basic Package reads from unit 1 (unless it is changed to meet computer requirements). It is permissible but unwise to use that unit for other major options.

IAPART--indicates whether array BUFF is separate from array RHS.

If IAPART = 0, the arrays BUFF and RHS occupy the same space. This option conserves space. This option should be used unless some other package explicitly says otherwise.

If IAPART ≠ 0, the arrays BUFF and RHS occupy different space. This option is not needed in the program as documented in this publication. It may be needed for packages yet to be written.

ISTRT--indicates whether starting heads are to be saved. If they are saved, they will be stored in array STRT. They must be saved if drawdown is calculated.

If ISTRT = 0, starting heads are not saved.

If ISTRT ≠ 0, starting heads are saved.

IBOUND--is the boundary array.

If IBOUND(I,J,K) < 0, cell I,J,K has a constant head.

If IBOUND(I,J,K) = 0, cell I,J,K is inactive (no-flow).

If IBOUND(I,J,K) > 0, cell I,J,K is variable-head.

HNOFLO--is the value of head to be assigned to all inactive cells (IBOUND = 0) throughout the simulation. Since heads at inactive cells are unused, this does not affect model results but serves to identify inactive cells when head is printed. This value is also used as drawdown at inactive cells if the drawdown option is used. Even if the user does not anticipate having inactive cells, a value for HNOFLO must be submitted.

Shead--is head at the start of the simulation. Regardless of whether starting head is saved, these values must be input to initialize the solution.

PERLEN--is the length of a stress period. It is specified for each stress period.

NSTP--is the number of time steps in a stress period.

TSMULT--is the multiplier for the length of successive time steps. The length of the first time step DELT(1) is related to PERLEN, NSTP and TSMULT by the relation

DELT(1) = PERLEN(1-TSMULT)/(1-TSMULT**NSTP).

SAMPLE INPUT TO THE BASIC PACKAGE(EXCLUDING OUTPUT CONTROL)

```
DATA          EXPLANATION                                                    INPUT RECORDS
ITEM

 1   {HEADNG}________________________________________________  SAMPLE INPUT FOR BASIC PACKAGE
 2   {HEADNG} CONTINUED______________________________________  WITH RECHARGE, DRAINS, AND WELLS
 3   {NLAY, NROW, NCOL, NPER, ITMUNI}________________________          3         10         12         2         1
 4   {IUNIT TABLE}___________________________________________   31 12 14 00 00 00 00 28 19 00 00 22
 5   {IAPART, ISTRT}_________________________________________          0         1
 6   CONTROL RECORD FOR IBOUND ARRAY LAYER 1_________________          1         1 (12I2)                    3
                                                               -1 1 1 1 1 1 1 1 1 1 1 1
                                                               -1 1 1 1 1 1 1 1 1 1 1 1
                                                               -1 1 1 1 1 1 1 1 1 1 1 1
                                                               -1 1 1 1 1 1 1 1 1 1 1 1
         IBOUND VALUES FOR LAYER 1                             -1 1 1 1 1 1 1 1 1 1 1 1
                                                               -1 1 1 1 1 1 1 1 1 1 1 1
                                                               -1 1 1 1 1 1 1 1 1 1 1 1
                                                               -1 1 1 1 1 1 1 1 1 1 1 1
                                                               -1 1 1 1 1 1 1 1 1 1 1 1
                                                               -1 1 1 1 1 1 1 1 1 1 1 1
 6   CONTROL RECORD FOR IBOUND ARRAY LAYER 2_________________          1         1 (12I3)                    3
                                                                -2  2  2  2  2  2  2  2  2  2  2  2
                                                                -2  2  2  2  2  2  2  2  2  2  2  2
                                                                -2  2  2  2  2  2  2  2  2  2  2  2
                                                                -2  2  2  2  2  2  2  2  2  2  2  2
         IBOUND VALUES FOR LAYER 2                              -2  2  2  2  2  2  2  2  2  2  2  2
                                                                -2  2  2  2  2  2  2  2  2  2  2  2
                                                                -2  2  2  2  2  2  2  2  2  2  2  2
                                                                -2  2  2  2  2  2  2  2  2  2  2  2
                                                                -2  2  2  2  2  2  2  2  2  2  2  2
                                                                -2  2  2  2  2  2  2  2  2  2  2  2
 6   CONTROL RECORD FOR IBOUND ARRAY LAYER 3_________________           0         3
 7   {HNOFLO}________________________________________________     999.99
 8   CONTROL RECORD FOR STARTING HEAD ARRAY FOR LAYER 1______          1         1 (12F5.1)                  6
                                                                10.  17.  24.  31.  38.  45.  52.  59.  66.  73.  80.  87.
                                                                11.  19.  27.  35.  43.  51.  59.  67.  75.  83.  91.  99.
                                                                13.  23.  33.  43.  53.  63.  73.  83.  93. 103. 113. 123.
                                                                16.  29.  42.  55.  68.  81.  94. 107. 120. 133. 146. 159.
     STARTING HEAD VALUES FOR LAYER 1                           20.  37.  54.  71.  88. 105. 122. 139. 156. 173. 190. 207.
                                                                20.  37.  54.  71.  88. 105. 122. 139. 156. 173. 190. 207.
                                                                16.  29.  42.  55.  68.  81.  94. 107. 120. 133. 146. 159.
                                                                13.  23.  33.  43.  53.  63.  73.  83.  93. 103. 113. 123.
                                                                11.  19.  27.  35.  43.  51.  59.  67.  75.  83.  91.  99.
                                                                10.  17.  24.  31.  38.  45.  52.  59.  66.  73.  80.  87.
 8   CONTROL RECORD FOR STARTING HEAD ARRAY FOR LAYER 2______         23         1 (12F6.2)                  5
 8   CONTROL RECORD FOR STARTING HEAD ARRAY FOR LAYER 3______          0       0.0
 9   STRESS PERIOD 1----{PERLEN  NSTP  TSMULT}_______________      86400         5      1.5
 9   STRESS PERIOD 2----{PERLEN  NSTP  TSMULT}_______________     172800         7      1.5
```

FIELDS IN ARRAY CONTROL RECORDS ARE---{ LOCAT, CONST, FMTIN, IPRN}

Output Control Input

Output Control is a major option separate from the rest of the Basic Package. Input to Output Control is read from the unit specified in IUNIT(12). If IUNIT(12) is zero, no output control data are read, and default output control is used. Under the default, head and total budget are printed at the end of every stress period. Additionally, if starting heads are saved (ISTRT is not 0), drawdown is printed at the end of every stress period. The default printout format for head and drawdown is 10G11.4. All printer output goes to unit 6 as specified in the main program. If necessary, the unit number for printer output can be changed to meet the requirements of a particular computer.

FOR EACH SIMULATION

BAS1RP

1.	Data:	IHEDFM	IDDNFM	IHEDUN	IDDNUN
	Format:	I10	I10	I10	I10

FOR EACH TIME STEP

BAS1OC

2.	Data:	INCODE	IHDDFL	IBUDFL	ICBCFL
	Format:	I10	I10	I10	I10
3.	Data:	Hdpr	Ddpr	Hdsv	Ddsv
	Format:	I10	I10	I10	I10

(Record 3 is read 0, 1, or NLAY times, depending on the value of INCODE.)

Explanation of Fields Used in Input Instructions

IHEDFM--is a code for the format in which heads will be printed.

IDDNFM--is a code for the format in which drawdowns will be printed. Format codes have the same meaning for both head and drawdown. A positive format code indicates that each row of data is printed completely before starting the next row. This means that when there are more columns in a row than will fit on one line, additional lines are used as required to complete the row. This format is called the wrap format. A negative format code indicates that the printout is broken into strips where only that number of columns that will fit across one line are printed in a strip. As many strips are used as are required to print the entire model width. This format is called the strip format. The absolute value of the format code specifies the printout format as follows.

0 - (10G11.4)	7 - (20F5.0)
1 - (11G10.3)	8 - (20F5.1)
2 - (9G13.6)	9 - (20F5.2)
3 - (15F7.1)	10 - (20F5.3)
4 - (15F7.2)	11 - (20F5.4)
5 - (15F7.3)	12 - (10G11.4)
6 - (15F7.4)	

IHEDUN--is the unit number to which heads will be written if they are saved on disk.

IDDNUN--is the unit number to which drawdowns will be written if they are saved on disk.

INCODE--is the head/drawdown ouput code. It determines the number of records in input item 3.

If INCODE < 0, layer-by-layer specifications from the last time steps are used. Input item 3 is not read.

If INCODE = 0, all layers are treated the same way. Input item 3 will consist of one record.

If INCODE > 0, input item 3 will consist of one record for each layer.

IHDDFL--is a head and drawdown output flag.

If IHDDFL = 0, neither heads nor drawdowns will be printed or saved on disk.

If IHDDFL ≠ 0, heads and drawdowns will be printed or saved according to the flags for each layer specified in input item 3.

IBUDFL--is a budget print flag.

If IBUDFL = 0, overall volumetric budget will not be printed.

If IBUDFL ≠ 0, overall volumetric budget will be printed.

(Note that the overall volumetric budget will always be printed at the end of a stress period, even if the value of IBUDFL is zero.)

ICBCFL--is a cell-by-cell flow-term flag.

If ICBCFL = 0, cell-by-cell flow terms are not saved or printed.

If ICBCFL ≠ 0, cell-by-cell flow terms are printed or recorded on disk depending on flags set in the component of flow packages, i.e., IWELCB, IRCHCB, etc.

Hdpr--is the output flag for head printout.

If Hdpr = 0, head is not printed for the corresponding layer.

If Hdpr ≠ 0, head is printed for the corresponding layer.

Ddpr--is the output flag for drawdown printout.

If Ddpr = 0, drawdown is not printed for the corresponding layer.

If Ddpr ≠ 0, drawdown is printed for the corresponding layer.

Hdsv--is the output flag for head save.

If Hdsv = 0, head is not saved for the corresponding layer.

If Hdsv ≠ 0, head is saved for the corresponding layer.

Ddsv--is the output flag for drawdown save.

If Ddsv = 0, drawdown is not saved for the corresponding layer.

If Ddsv ≠ 0, drawdown is saved for the corresponding layer.

SAMPLE INPUT TO THE OUTPUT CONTROL OPTION

DATA ITEM	EXPLANATION	INPUT RECORDS			
1	{IHEDFM, IDDNFM, IHEDUN, IDDNUN}	4	8	76	77
2	TIME STEP 1--{INCODE, IHDDFL, IBUDFL, ICBCFL}	1	1	0	0
3	LAYER 1--{HDPR, DDPR, HDSV, DDSV}	1	1	1	1
3	LAYER 2--{HDPR, DDPR, HDSV, DDSV}	1	1	0	0
3	LAYER 3--{HDPR, DDPR, HDSV, DDSV}	1	1	0	0
2	TIME STEP 2--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	0	1	0
2	TIME STEP 3--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	1	1	0
2	TIME STEP 4--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	0	1	0
2	TIME STEP 5--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	1	1	1
2	TIME STEP 6--{INCODE, IHDDFL, IBUDFL, ICBCFL}	0	1	1	0
3	ALL LAYERS--{HDPR, DDPR, HDSV, DDSV}	1	1	0	0
2	TIME STEP 7--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	0	1	0
2	TIME STEP 8--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	1	1	0
2	TIME STEP 9--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	0	1	0
2	TIME STEP 10--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	1	1	0
2	TIME STEP 11--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	0	1	0
2	TIME STEP 12--{INCODE, IHDDFL, IBUDFL, ICBCFL}	-1	1	1	1